Homework No. 3 - Solution

```
> restart; with(student);
```

1. For the two soils whose characteristics are given below, and for the variable diffusivity case:

a) Compute the infiltration as a function of time for constant $\theta_1 = n$ at the surface.

b) Assuming a rainfall duration of 2 hours, calculate the maximum depth of penetration of an infiltration moisture wave for the two soils above.

c) Compute the time to ponding , t_o, and volume of surface rainfall excess, Rs*, for the following conditions:
Fine Sand: Rain intensity: i = 5 cm/h, and rain duration, tr = 1.5 h.
Light Clay: Rain intensity: i = 0.5 cm/h, and rain duration, tr = 1.5 h.

Define input parameters:
Fine Sand:
Porosity and pore disconnectedness index:

```
> n := 0.3; c := 4.22;
```

$$n := 0.3$$

$$c := 4.22$$

Pore size distribution index:

```
> m := 1.36;
```

$$m := 1.36$$

Diffusivity index, d:

```
> d := c-(1/m)-1;
```

$$d := 2.484705882$$

Saturated Hydraulic Conductivity (cm/s)

```
> K(1):=0.001;
```

$$K(1) := 0.001$$

Soil Matric Capillary Potential (cm):

```
> psi(1):=24;
```

$$\psi(1) := 24$$

Intial degree of saturation:

```
> so:=0.15/n;
```

$$so := 0.5000000000$$

Define Required Infiltration Functions:

Define dimensionless infiltration diffusivity (this expression is from Entekhabi, (1988).)

```
> phi:=(m,s)->3*Pi/(10*(1-s)^2)*(m/(1+4*m)+m^2*s^(4+1/m)/(1+4*m)/
```

```
(1+3*m)-m*s/(1+3*m));
```

$$\phi := (m, s) \rightarrow \frac{3}{10} \frac{\pi \left(\frac{m}{1+4m} + \frac{m^2 s^{4+\frac{1}{m}}}{(1+4m)(1+3m)} - \frac{m s}{1+3m} \right)}{(1-s)^2}$$

```
> evalf(phi(m,so));
```

0.2994984441

Define variable diffusivity function, D_i (cm/sqrt(s)):

```
> Difv:=s->5*K(1)*psi(1)*phi(m,s)/m/n/3;
```

$$Difv := s \rightarrow \frac{5}{3} \frac{K(1)\,\psi(1)\,\phi(m, s)}{m\,n}$$

```
> evalf(Difv(so));
```

0.02936259256

Define Sorptivity function, S_i (cm/sqrt(s))

```
> si := s->2*(1-s)*sqrt(5*n*K(1)*psi(1)*phi(m,s)/3/m/Pi);
```

$$si := s \rightarrow 2\,(1-s)\sqrt{\frac{5}{3} \frac{n\,K(1)\,\psi(1)\,\phi(m, s)}{m\,\pi}}$$

```
> evalf(si(so));
```

0.02900303976

Define gravitational infiltration A_o (cm/s)

```
> Ao:=s->K(1)*(1+s^c)/2;
```

$$Ao := s \rightarrow \frac{1}{2} K(1)\left(1 + s^c\right)$$

```
> evalf(Ao(so));
```

0.0005268301700

Define Infiltration Capacity Function (Eagleson, 1978):

```
> f:=(t,s)->1/2*si(s)*t^(-1/2)+Ao(s);
```

$$f := (t, s) \rightarrow \frac{1}{2} \frac{si(s)}{\sqrt{t}} + Ao(s)$$

Define procedure "inf" to compute and plot the infiltration capacity as a function of time:

```
> inf:=proc() global so: local t: for t from 100. by 100. while t
  <=1000. do printf("%g %g\n", t, evalf(f(t,so))) od: plot(f(tt,
  so),tt=100..36000,axes=boxed,labels=["t(sec)","fi(t) (cm/s)"]):
  end;
```

inf := **proc**( )

local *t*;

```
    global so;
    for t from 100. by 100. while t <= 1000. do
        printf("%g %g\n", t, evalf(f(t, so)))
    end do;
    plot(f(tt, so), tt = 100 ..36000, axes = boxed, labels = ["t(sec)", "fi(t) (cm/s)"])
end proc
```

```
> inf();
```

```
100 0.00197698
200 0.00155224
300 0.00136408
400 0.00125191
500 0.00117536
600 0.00111885
700 0.00107494
800 0.00103954
900 0.00101021
1000 0.000985408
```




```
> tr:=2*3600;
```

$tr := 7200$

Define Diffusive depth of penetration function:

```
> zmaxdif:=(tr,s)->4*(Difv(s)*tr)^(1/2);
```

$$zmaxdif := (tr, s) \rightarrow 4\sqrt{Difv(s)\ tr}$$

Evaluate diffusive depth of penetration (cm):

```
> evalf(zmaxdif(tr,so));
```

$$58.15987159$$

Define gravitational depth of penetration function:

```
> zmaxgrav:=tr->K(1)*so^c/n*tr;
```

$$zmaxgrav := tr \rightarrow \frac{K(1)\ so^c\ tr}{n}$$

Evaluate gravitational depth of penetration (cm):

```
> evalf(zmaxgrav(tr));
```

$$1.287848155$$

Define total maximum depth of penetration function:

```
> zmax:=(tr,s)->4*(Difv(s)*tr)^(1/2)+K(1)*so^c/n*tr;
```

$$zmax := (tr, s) \rightarrow 4\sqrt{Difv(s)\ tr} + \frac{K(1)\ so^c\ tr}{n}$$

Evaluate total (maximum) depth of penetration (cm):

```
> evalf(zmax(tr,so));
```

$$59.44771974$$

Define time to ponding function (i >> Ao):

```
> tto:=(ir,s)->(si(s))^2/2/(ir-Ao(s))^2;
```

$$tto := (ir, s) \rightarrow \frac{1}{2}\frac{si(s)^2}{(ir - Ao(s))^2}$$

Define time to ponding function (i > Ao):

```
> ttof:=(ir,s)->(si(s))^2/2/(ir-Ao(s))^2*(1-Ao(s)/2/ir);
```

$$ttof := (ir, s) \rightarrow \frac{1}{2}\frac{si(s)^2\left(1 - \frac{1}{2}\frac{Ao(s)}{ir}\right)}{(ir - Ao(s))^2}$$

Time Compression Approximation displacement t' (for the case of i_r >> Ao):

```
> tp:=(ir,s)->fsolve(tto(ir,s)-tt-(si(s))^2/4/(ir-Ao(s))^2=0,tt);
```

$$tp := (ir, s) \rightarrow fsolve\left(tto(ir, s) - tt - \frac{1}{4}\frac{si(s)^2}{(ir - Ao(s))^2} = 0, tt\right)$$

Time Compression Approximation displacement t':

```
> tpp:=(ir,s)->fsolve(ttof(ir,s)-tt-(si(s))^2/4/(ir-Ao(s))^2=0,tt)
  ;
```

$$tpp := (ir, s) \rightarrow fsolve\left(ttof(ir, s) - tt - \frac{1}{4} \frac{si(s)^2}{(ir - Ao(s))^2} = 0, tt \right)$$

c) Compute time to ponding, t_o (seconds) for new rainfall intensity, i_r, and new rainfall duration, t_r . i_r in cm/s and t_r in seconds.

```
> ir:=evalf(5/3600); tr:=1.5*3600;
```

$$ir := 0.001388888889$$

$$tr := 5400.0$$

Evaluate time to ponding, to, and t' in seconds assuming i_r >>A_o:

```
> tto(ir,so);
```

565.9568785

```
> tp(ir,so);
```

282.9784393

However, in this case, i_r is not much larger that A_o:

```
> ttof(ir,so);
```

458.6181414

```
> tpp(ir,so);
```

175.6397022

Define Surface Retention Capacity,ho, and t*,

```
> ho:=0.;ho/ir;
```

$$ho := 0.$$

0.

Define rainfall excess volume (ir >> Ao and negligible ho):

```
> Rsf:=(s,ir,tr)->(ir-Ao(s))*tr+Ao(s)*tp(ir,s)-si(s)*sqrt(tr-tp
(ir,s));
```

$$Rsf := (s, ir, tr) \rightarrow (ir - Ao(s))\, tr + Ao(s)\, tp(ir, s) - si(s) \sqrt{tr - tp(ir, s)}$$

```
> evalf(Rsf(so,ir,tr));
```

2.729513784

Define rainfall excess volume (ir >> Ao and ho):

```
> Rsfh:=(s,ir,tr)->(ir-Ao(s))*tr+Ao(s)*(ho/ir+tp(ir,s))-si(s)*sqrt
(tr-(ho/ir+tp(ir,s)));
```

$$Rsfh := (s, ir, tr) \rightarrow (ir - Ao(s))\, tr + Ao(s) \left(\frac{ho}{ir} + tp(ir, s) \right)$$

$$- si(s) \sqrt{tr - \frac{ho}{ir} - tp(ir, s)}$$

```
> evalf(Rsfh(so,ir,tr));
```

2.729513784

Relative difference (%):

```
> (1-Rsf(so,ir,tr)/Rsfh(so,ir,tr))*100;
```

$$0.$$

Define rainfall excess volume (ir > Ao and negligible ho):

```
> Rsff:=(s,ir,tr)->(ir-Ao(s))*tr+Ao(s)*tpp(ir,s)-si(s)*sqrt(tr-tpp(ir,s));
```

$$Rsff := (s, ir, tr) \rightarrow (ir - Ao(s))\, tr + Ao(s)\, tpp(ir, s) - si(s) \sqrt{tr - tpp(ir, s)}$$

```
> evalf(Rsff(so,ir,tr));
```

$$2.651317307$$

Define rainfall excess volume (ir > Ao and ho):

```
> Rsffh:=(s,ir,tr)->(ir-Ao(s))*tr+Ao(s)*(ho/ir+tpp(ir,s))-si(s)*sqrt(tr-(ho/ir+tpp(ir,s)));
```

$$Rsffh := (s, ir, tr) \rightarrow (ir - Ao(s))\, tr + Ao(s) \left(\frac{ho}{ir} + tpp(ir, s) \right) - si(s) \sqrt{tr - \frac{ho}{ir} - tpp(ir, s)}$$

```
> evalf(Rsffh(so,ir,tr));
```

$$2.651317307$$

Relative difference (%):

```
> (1-Rsff(so,ir,tr)/Rsffh(so,ir,tr))*100;
```

$$0.$$

Define volume of rainfall excess as the integral of the precipitation excess function (negligible effect of ho on infiltration dynamics):

```
> Rss:=(ir,s)->int(ir-f(t-tpp(ir,s),s),t=ttof(ir,s)..tr);
```

$$Rss := (ir, s) \rightarrow \int_{ttof(ir, s)}^{tr} (ir - f(t - tpp(ir, s), s))\, \mathrm{d}t$$

```
> evalf(Rss(ir,so));
```

$$2.651317307$$

Define volume of rainfall excess as the integral of the precipitation excess function (and ho):

```
> Rssh:=(ir,s)->ho+int(ir-f(t-(ho/ir+tpp(ir,s)),s),t=ho/ir+ttof(ir,s)..tr);
```

$$Rssh := (ir, s) \rightarrow ho + \int_{\frac{ho}{ir} + ttof(ir, s)}^{tr} \left(ir - f\left(t - \frac{ho}{ir} - tpp(ir, s), s \right) \right) \mathrm{d}t$$

```
> evalf(Rssh(ir,so));
```

$$2.651317307$$

```
> irain:=t->piecewise(t<=tr, ir,t>tr,0); iint:=t->piecewise(t<=ho/ir, ir,t>ho/ir,0);
```

$$irain := t \rightarrow piecewise(t \le tr, ir, tr < t, 0)$$

$$iint := t \rightarrow piecewise\left(t \le \frac{ho}{ir}, ir, \frac{ho}{ir} < t, 0 \right)$$

```
> plot([f(t-(ho/ir+tpp(ir,so)),so),irain(t),iint(t)],t=1...6000,v=
0..0.006,axes=boxed,labels=["t(s)","Rates (cm/s)"],color=[red,
blue,magenta]);
```



Repeat analysis for a new soil:

Light Clay:

```
> n := 0.48; c := 8.54;
```

$$n := 0.48$$

$$c := 8.54$$

```
> m := 0.22;
```

$$m := 0.22$$

```
> d := c-1/m-1;
```

$$d := 2.994545455$$

```
> K(1):=0.00001;
```

$$K(1) := 0.00001$$

```
> psi(1):=26;
```

$$\psi(1) := 26$$

```
> so:=0.24/n;
```

$$so := 0.4999999999$$

```
> evalf(phi(m,so));
```

0.1915029125

```
> evalf(Difv(so));
```

0.0007858389720

```
> evalf(si(so));
```

0.007591595898

```
> evalf(Ao(so));
```

0.000005013433025

```
> inf();
```

```
100 0.000384593
200 0.000273417
300 0.000224164
400 0.000194803
500 0.000174767
600 0.000159976
700 0.000148481
800 0.000139215
900 0.00013154
1000 0.000125047
```




```
> tr:=2*3600;
```

$tr := 7200$

```
> evalf(zmax(tr,so));
```

9.515057456

```
> ir:=0.5/3600; tr:=1.5*3600;
```

$ir := 0.0001388888889$

$tr := 5400.0$

```
> tto(ir,so);
```

1607.808216

```
> ttof(ir,so);
```

1578.789917

```
> evalf(Rsf(so,ir,tr));evalf(Rsfh(so,ir,tr));
```

0.2122892401

0.2122892401

```
> evalf(Rsff(so,ir,tr));evalf(Rsffh(so,ir,tr));
```

0.2105215879

0.2105215879

```
> (1-Rsff(so,ir,tr)/Rsffh(so,ir,tr))*100;
```

0.

```
> evalf(Rss(ir,so));evalf(Rssh(ir,so));ho/ir;
```

0.2105215878

0.2105215878

0.

```
> plot([f(t-(ho/ir+tpp(ir,so)),so),irain(t),iint(t)],t=.1...6000,
  v=0...0.00020,axes=boxed,labels=["t(s)","Rates (cm/s)"],color=
  [red,blue,magenta]);
```



2. Assume that rainfalls are of constant intensity i r and duration t r . Infiltration capacity is given by Philip's equation evaluated at the initial soil moisture saturation before the storm, so. That is,

$$f := (t, s) \rightarrow \frac{1\, si(s)}{2\sqrt{t}} + Ao(s)$$

If no evaporation occurs during the storm, show that the degree of soil moisture saturation at the end of the storm s f can be approximated by,

$$sf := \begin{cases} so + \dfrac{ir\, tr - K(1)\, so^{c}\, tr}{n\, z} & tr - tto(ir, so) \leq 0 \textbf{ or } ir - Ao(so) \leq 0 \\ so + \dfrac{1}{n\, z}\Big(ir\, tto(ir, so) + si(so)\left(\sqrt{tr} - \sqrt{tto(ir, so)}\right) + Ao(so)\,(tr - tto(ir, so)) - K(1)\, so^{c}\, tr\Big) & otherwise \end{cases}$$

where z is a characteristic soil depth over which the mass balance should be performed. Ao is the gravitational infiltration rate, n is the porosity, Si is the sorptivity,tto is the ponding time, K(1) is the saturated hydraulic conductivity, and c is the pore disconnectedness index.

Use the expressions you derived and compute the final degree of saturation for the two soils of Problem 1 and for two storms of intensity equal to 0.6 cm/h and whose duration are equal to 2to/3 and 2to. State all your assumptions. Assume that the depth of the soil column over which the water balance is to be performed is 1 m. If neither of these rainfall durations is appropriate, assume a duration of 5 hours.

Define function for the soil moisture content at the end of a rainfall of intensity ir, duration tr.

```
> sf:=(ir,tr,so)->piecewise(tr<=tto(ir,so) or ir<=Ao(so),so+(ir*tr-K(1)*so^c*tr)/n/z,so+(ir*tto(ir,so)+si(so)*(tr^(1/2)-tto(ir,so)^(1/2))+Ao(so)(tr-tto(ir,so))-K(1)*so^c*tr)/n/z);
```

$$sf := (ir, tr, so) \rightarrow piecewise\left(tr \leq tto(ir, so) \textbf{ or } ir \leq Ao(so), so + \frac{tr\, ir - K(1)\, so^c\, tr}{n\, z}, so + \frac{1}{n\, z}\left(ir\, tto(ir, so) + si(so)\left(\sqrt{tr} - \sqrt{tto(ir, so)}\right) + Ao(so)\,(tr - tto(ir, so)) - K(1)\, so^c\, tr\right)\right)$$

Fine sand soil:

Compute gravitational infiltration rate (cm/s) and time to ponding (s) for the given intensity (cm/s):

```
> ir:=0.6/3600;
```

$$ir := 0.0001666666667$$

```
> Ao(so);
```

$$0.000005013433025$$

Observe that for the soil and rainfall characteristics under consideration, i r < Ao, so that ponding will not occur. Thus, the time to ponding, to, is not defined.

Define soil depth, z, in cm;

```
> z:=100;
```

$$z := 100$$

Compute the final degree of saturation for the given conditions. Since for this soil, there is no time to ponding (i.e., time to ponding is not defined) because ir < Ao, assume an arbitrary rainfall duration equal to 5 hours.

```
> tr:=5*3600;evalf(sf(ir,tr,so));
```

$$tr := 18000$$

$$0.5197860829$$

Light clay soil:

Compute gravitational infiltration rate (cm/s) and time to ponding (s) for the given intensity (cm/s)

```
> ir:=0.6/3600;
```

$$ir := 0.0001666666667$$

```
> Ao(so);
```

$$0.000005013433025$$

```
> tto(ir,so);
```

$$1102.725394$$

Define soil column depth (cm):

```
> z:=100;
```

$$z := 100$$

Compute the final degree of saturation for the given conditions and for tr = 2to/3 and tr=2to.

```
> sf(ir,2*tto(ir,so)/3,so); irain:=t->piecewise(t<=2*tto(ir,so)/3,
  ir,0); plot([f(t-tto(ir,so)/2,so),irain(t)],t=.1...1000,v=0..
  .0.001,axes=boxed,labels=["t(s)","Rates (cm/s)"],color=[red,
  blue]);
```

$$0.5025521935$$

$$irain := t \rightarrow piecewise\left(t \leq \frac{2}{3}\, tto(ir, so), ir, 0\right)$$




```
> sf(ir,2*tto(ir,so),so);irain:=t->piecewise(t<=2*tto(ir,so), ir,
  0); plot([f(t-tto(ir,so)/2,so),irain(t)],t=.1...3000,v=0..
  .0.001,axes=boxed,labels=["t(s)","Rates (cm/s)"],color=[red,
  blue]);
```

$$0.5060032313$$

$$irain := t \rightarrow piecewise(t \leq 2\ tto(ir, so), ir, 0)$$



Note that the above expressions, which have been taken from "Hydrology, An Introduction to Hydrologic Science" by R.L. Bras, ignore the effect of t' on the dynamics of the infiltration capacity. If this effect is not ignored the correct expressions would be:

$$s_f := \begin{cases} s_o + \dfrac{i_r t_r - K(1)\, s_o^c t_r}{n z} & t_r - t_o \leq 0 \textbf{ or } i_r - A_o \leq 0 \\ s_o + \dfrac{S_i \sqrt{t_r - \frac{1}{2} t_o} + A_o\left(t_r - \frac{1}{2} t_o\right) - K(1)\, s_o^c t_r}{n z} & \textit{otherwise} \end{cases}$$

3. Calculate the soil moisture flux q (cm/day) between depths ϕ_1 = -0.8 m and ϕ_2 = -1.8 m in a soil column. The corresponding piezometric heads are ψ_1 =-145 cm and ψ_2 = -230 cm, respectively. For

this soil the relationship between hydraulic conductivity and soil matric potential is given by,

$$K := \psi \rightarrow \frac{250 \cdot 1}{(-\psi)^{2.11}}$$

where K is in cm/day and is in cm. Note that a vertical coordinate positive in the upward direction is used.

```
> restart;with(student);
```

Define input data:

```
> phi[1]:=-145;phi[2]:=-230;z[1]:=-80;z[2]:=-180;
```

$$\phi_1 := -145$$

$$\phi_2 := -230$$

$$z_1 := -80$$

$$z_2 := -180$$

Define soil matrix potential function: $\psi := (\phi, z) \rightarrow \phi - z$

```
> psi0:=(phi,z)->phi-z;
```

$$\psi 0 := (\phi, z) \rightarrow \phi - z$$

Compute soil matrix potential at soil depths z_1 and z_2.

```
> psi[1]:=psi0(phi[1],z[1]);
```

$$\psi_1 := -65$$

```
> psi[2]:=psi0(phi[2],z[2]);
```

$$\psi_2 := -50$$

Compute gradient of piezometric head, $\frac{d}{dz}\phi(z)$, using a finite difference approximation:

```
> dfdz:=(phi[1]-phi[2])/(z[1]-z[2]);
```

$$dfdz := \frac{17}{20}$$

Define hydraulic conductivity function given in problem statement

```
> K:=psi->250*1/((-psi)^2.11);
```

$$K := \psi \rightarrow \frac{250}{(-\psi)^{2.11}}$$

Compute hydraulic conductivity at depths z_1 and z_2:

```
> K(psi[1]); K(psi[2]);
```

$$0.03738460758$$

$$0.06502993195$$

Define an (arithmetic) average hydraulic conductivity over the depth (z_2 - z_1):

```
> Kave:=(K(psi[1])+K(psi[2]))/2;
```

$$Kave := 0.05120726977$$

Define the average flux over the soil depth from z_1 to z_2 using Darcy's equation,

$q := -K(\psi)\left(\frac{d}{dz}\phi(z)\right)$. q is in cm/day.

```
> q:=-Kave*dfdz;
```

$$q := -0.04352617930$$

```
>
```